Derivative of the weighted mean invasion probability ($\bar{\pi}$) at recombination rate $r = 0$: generic expressions

Paths

In[142]:=
```
figPath := "/Users/Simon/Documents/LocAdD/results/130606/nonZeroOptRecRate/figures/"
```

Rules and assumptions

In[143]:=
```
R1Rule := R1 → √(-4 b (-1 + a + b) m (1 + m) qC + (b + (-1 + a) m + 2 b m qC)^2)
```

In[144]:=
```
qEqPCRule := {qEq → (b - m + a m + 2 b m qC + R1) / (2 (+b + b m))}
```

Check:

```
FullSimplify[qEq /. qEqPCRule /. R1Rule /. qC → 0,
 Assumptions → Flatten[{assumeGeneral, m < b/(1 - a)}]]
```

$$\frac{b - m + a\,m}{b + b\,m}$$

```
qEq /. qEqRule
```

$$\frac{b - m + a\,m}{b\,(1 + m)}$$

This is as expected.

Generic fitnesses

Recapitulation: Deterministic dynamics of the marginal one-locus system

- **Migration-selection equilibrium**
- Coordinates

We obtain the coordinates of the polymorphic one-locus migration-selection equilibrium as the solution of the recursion equation such that $0 < \hat{q}_B < 1$.

In[145]:=
```
eqqB := qB == (1 - m) w1Tilde/wBarTilde qB + m qC
```

```
In[146]:= w1TildeRule := w1Tilde → w33 qB + w34 (1 - qB)
          w2TildeRule := w2Tilde → w34 qB + w44 (1 - qB)
          wBarTildeRule := wBarTilde → qB^2 w33 + 2 qB (1 - qB) w34 + (1 - qB)^2 w44
```

Test:

```
qB w1Tilde + (1 - qB) w2Tilde - wBarTilde /.
  {w1TildeRule, w2TildeRule, wBarTildeRule} // Simplify

0
```

For a polymorphic continent, we obtain

```
In[149]:= qEqBRulePolymCont = FullSimplify[Solve[eqqB /. w1TildeRule, qB]][[1]]
```

$$\text{Out[149]= } \left\{qB \to \frac{1}{2\,(-1+m)\,(w33-w34)}\left(w34 - m\,w34 - wBarTilde + \sqrt{\left(4\,(-1+m)\,m\,qC\,(w33-w34)\,wBarTilde + \left((-1+m)\,w34 + wBarTilde\right)^2\right)}\right)\right\}$$

An non-trivial explicit solution for $\hat{q}_B$ cannot be found by *Mathematica*:

FullSimplify[Solve[eqqB /. w1TildeRule /. wBarTildeRule, qB], Assumptions -> {0 < w33, 0 < w34, 0 < w44, 0 < qB < 1, 0 < qC < 1, 0 < m < 1}]

```
$Aborted
```

For a monomorphic continent, we find

```
In[150]:= qEqBRuleRaw = FullSimplify[Solve[(eqqB /. w1TildeRule /. {qC → 0}), qB]]
```

$$\text{Out[150]= } \left\{\{qB \to 0\},\ \left\{qB \to \frac{w34 - m\,w34 - wBarTilde}{(-1+m)\,(w33-w34)}\right\}\right\}$$

```
In[151]:= qEqBRule := qEqBRuleRaw[[2]]
```

- Stability

We find the condition for stability of $\hat{q}_B$ by asking when the derivative of the right-hand side of the difference equation is > 0.

$$\text{In[152]:= } \mathbf{DeqqB := D\left[(1-m)\,\frac{w1Tilde}{wBarTilde}\,qB + m\,qC - qB \;/.\; w1TildeRule,\ qB\right]}$$

```
DeqqB
```

$$\text{Out[153]= } -1 + \frac{(1-m)\,qB\,(w33-w34)}{wBarTilde} + \frac{(1-m)\,(qB\,w33 + (1-qB)\,w34)}{wBarTilde}$$

```
DeqqB /. qEqBRule
```

$$-1 + \frac{(1-m)\left(w34\left(1 - \frac{w34-m\,w34-wBarTilde}{(-1+m)\,(w33-w34)}\right) + \frac{w33\,(w34-m\,w34-wBarTilde)}{(-1+m)\,(w33-w34)}\right)}{wBarTilde} + \frac{(1-m)\,(w34-m\,w34-wBarTilde)}{(-1+m)\,wBarTilde}$$

```
mCritRule = Solve[DeqqB == 0 /. qEqBRule, m][[1]]
```

$$\left\{m \to \frac{w34 - wBarTilde}{w34}\right\}$$

```
Simplify[Reduce[DeqqB < 0 /. qEqBRule, m], Assumptions → {0 < w34, 0 < wBarTilde, 0 < w33}]

m w34 + wBarTilde < w34
```

We note that $\tilde{\bar{w}}$ is a function of q_B and therefore, the generic expressions obtained above for the equilibrium frequency and the critical migration rate are not very informative; they provide implicit solutions only. To obtain informative explicit solutions, we make specific assumptions about the relative fitnesses. For details, see *Mathematica* Notebook '2LocContIsland_Determ_Discr.nb '. Here, however, we want to proceed with generic implicit expressions, as our goal is to obtain a generic condition for when the recombination rate at which the invasion probability of the focal mutation A_1 at a second locus has a maximum is strictly positive.

A generic (implicit) condition for $r_{\mathbf{opt}} > 0$

Probability-generating functions

We start from the most general versions of the probability-generating functions (pgf) for the two-type branching process under the assumption of type-specific, but independent Poisson-distributed offspring numbers for each parental type. These are

$$f_i(s_1, s_2) = e^{-\lambda_{i1}(1-s_1)}\, e^{-\lambda_{i2}(1-s_2)} \text{ for } i \in \{1, 2\}, \tag{1}$$

with

$$\lambda_{11} = (1-m)[w_1 - r(1-\hat{q}_B)\, w_{14}]/\overline{w} \tag{2}$$

$$\lambda_{12} = (1-m)\, r(1-\hat{q}_B)\, w_{14}/\overline{w} \tag{3}$$

$$\lambda_{21} = (1-m)\, r\, \hat{q}_B\, w_{14}/\overline{w} \tag{4}$$

$$\lambda_{22} = (1-m)[w_2 - r\, \hat{q}_B\, w_{14}]/\overline{w} \tag{5}$$

(see sections 2 and 4 of Text S1 of the manuscript).

We denote the extinction probabilities of allele A_1 conditional on occurrence on background B_1 (B_2) by Q_1 (Q_2). These are obtained as the smalles solution between 0 and 1 to

$$s_i = f_i(s_1, s_2) \text{ for } i \in \{1, 2\}. \tag{6}$$

The respective invasion probabilities of A_1 are π_1 and π_2 and the average invasion probability is obtained as the weighted mean

$$\overline{\pi} = \hat{q}_B\, \pi_1 + (1-\hat{q}_B)\, \pi_2. \tag{7}$$

As Eqs. (6) are transcendental, explicit solutions are not available in general.

```
In[154]:= λ11Rule := λ11 → (1 - m) (w1 - r (1 - qEqB) w14) / wBar
λ12Rule := λ12 → (1 - m) r (1 - qEqB) w14 / wBar
λ21Rule := λ21 → (1 - m) r qEqB w14 / wBar
λ22Rule := λ22 → (1 - m) (w2 - r qEqB w14) / wBar
λRules := {λ11Rule, λ12Rule, λ21Rule, λ22Rule}
```

```
In[159]:= w1Rule := w1 → w13 qEqB + w14 (1 - qEqB)
w2Rule := w2 → w24 (1 - qEqB) + w14 qEqB
wBarRule := wBar → qEqB^2 w33 + 2 qEqB (1 - qEqB) w34 + (1 - qEqB)^2 w44
```

Implicit differentiation at $r = 0$

Goal

We want to find the derivative of the average invasion probability as a function of the recombination rate, evaluated at $r = 0$:

$$\overline{\pi}'(0) = \frac{d}{dr}[\hat{q}_B\, \pi_1(r) + (1-\hat{q}_B)\, \pi_2(r)]\,|_{r=0} = \hat{q}_B\, \frac{d\pi_1(r)}{dr}\,|_{r=0} + (1-\hat{q}_B)\, \frac{d\pi_2(r)}{dr}\,|_{r=0}\,. \tag{8}$$

The optimal recombination rate is non-zero if $\overline{\pi}'(0)$ is positive. We obtain $\overline{\pi}'(0)$ via implicit differentiation. Let Q_1° and Q_2° be the smallest positive solutions to Eq. (6) with $r = 0$, and let $\pi_1^\circ = 1 - Q_1^\circ$ and $\pi_2^\circ = 1 - Q_2^\circ$ be the corresponding invasion probabilities in the absence of recombination.

The extinction probabilities Q_i fulfill Eq. (6). Moreover, because $Q_i = 1 - \pi_i$, we have $\frac{dQ_i(r)}{dr} = -\frac{d\pi_i(r)}{dr}$.

Implementation 1: in terms of the invasion probabilities π_i

```
In[162]:= eq1LHS[π1_, π2_] := Log[1 - π1]
eq1RHS[π1_, π2_] := -λ11 π1 - λ12 π2
eq2LHS[π1_, π2_] := Log[1 - π2]
eq2RHS[π1_, π2_] := -λ21 π1 - λ22 π2

eq1LHS[π1[r], π2[r]] == eq1RHS[π1[r], π2[r]] /. λRules
```

$$\text{Log}[1-\pi1[r]] == -\frac{(1-m)\,(w1-(1-qEqB)\, r\, w14)\, \pi1[r]}{wBar} - \frac{(1-m)\,(1-qEqB)\, r\, w14\, \pi2[r]}{wBar}$$

```
eq2LHS[π1[r], π2[r]] == eq2RHS[π1[r], π2[r]] /. λRules
```

$$\text{Log}[1-\pi2[r]] == -\frac{(1-m)\, qEqB\, r\, w14\, \pi1[r]}{wBar} - \frac{(1-m)\,(-qEqB\, r\, w14 + w2)\, \pi2[r]}{wBar}$$

```
D[eq1LHS[π1[r], π2[r]] /. λRules, r] == D[eq1RHS[π1[r], π2[r]] /. λRules, r]
```

$$-\frac{\pi 1'[r]}{1-\pi 1[r]} == \frac{(1-m)(1-qEqB)\,w14\,\pi 1[r]}{wBar} - \frac{(1-m)(1-qEqB)\,w14\,\pi 2[r]}{wBar} - \frac{(1-m)(w1-(1-qEqB)\,r\,w14)\,\pi 1'[r]}{wBar} - \frac{(1-m)(1-qEqB)\,r\,w14\,\pi 2'[r]}{wBar}$$

```
D[eq2LHS[π1[r], π2[r]] /. λRules, r] == D[eq2RHS[π1[r], π2[r]] /. λRules, r]
```

$$-\frac{\pi 2'[r]}{1-\pi 2[r]} == -\frac{(1-m)\,qEqB\,w14\,\pi 1[r]}{wBar} + \frac{(1-m)\,qEqB\,w14\,\pi 2[r]}{wBar} - \frac{(1-m)\,qEqB\,r\,w14\,\pi 1'[r]}{wBar} - \frac{(1-m)(-qEqB\,r\,w14+w2)\,\pi 2'[r]}{wBar}$$

In[166]:=
```
Deq1 = FullSimplify[
  D[eq1LHS[π1[r], π2[r]] /. λRules, r] == D[eq1RHS[π1[r], π2[r]] /. λRules, r]]
```

$$\frac{\pi 1'[r]}{-1+\pi 1[r]} == \frac{1}{wBar}(-1+m)\,(w1\,\pi 1'[r] + (-1+qEqB)\,w14\,(\pi 1[r]-\pi 2[r]+r\,(\pi 1'[r]-\pi 2'[r])))$$

```
Deq2 = FullSimplify[
  D[eq2LHS[π1[r], π2[r]] /. λRules, r] == D[eq2RHS[π1[r], π2[r]] /. λRules, r]]
```

$$\frac{\pi 2'[r]}{-1+\pi 2[r]} == \frac{1}{wBar}(-1+m)\,(qEqB\,w14\,(\pi 1[r]-\pi 2[r]+r\,\pi 1'[r]) + (-qEqB\,r\,w14+w2)\,\pi 2'[r])$$

```
Deq1 /. {r → 0}
```

$$\frac{\pi 1'[0]}{-1+\pi 1[0]} == \frac{(-1+m)\,((-1+qEqB)\,w14\,(\pi 1[0]-\pi 2[0]) + w1\,\pi 1'[0])}{wBar}$$

```
Deq2 /. {r → 0}
```

$$\frac{\pi 2'[0]}{-1+\pi 2[0]} == \frac{(-1+m)\,(qEqB\,w14\,(\pi 1[0]-\pi 2[0]) + w2\,\pi 2'[0])}{wBar}$$

```
Deq1R0 = Deq1 /. {r → 0} /. {π1[0] → π1Circ, π2[0] → π2Circ}
```

$$\frac{\pi 1'[0]}{-1+\pi 1Circ} == \frac{(-1+m)\,((-1+qEqB)\,w14\,(\pi 1Circ-\pi 2Circ) + w1\,\pi 1'[0])}{wBar}$$

```
Deq2R0 = Deq2 /. {r → 0} /. {π1[0] → π1Circ, π2[0] → π2Circ}
```

$$\frac{\pi 2'[0]}{-1+\pi 2Circ} == \frac{(-1+m)\,(qEqB\,w14\,(\pi 1Circ-\pi 2Circ) + w2\,\pi 2'[0])}{wBar}$$

Now we solve for $\pi_1'[0]$ and $\pi_2'[0]$:

```
πiDR0Rule = Solve[{Deq1R0, Deq2R0}, {π1'[0], π2'[0]}]
```

$$\left\{\left\{\pi 1'[0] \to -\frac{(-1+m)(-1+qEqB)\,w14\,(-1+\pi 1Circ)(\pi 1Circ-\pi 2Circ)}{w1-m\,w1-wBar-w1\,\pi 1Circ+m\,w1\,\pi 1Circ},\ \pi 2'[0] \to \frac{(-1+m)\,qEqB\,w14\,(-1+\pi 2Circ)(-\pi 1Circ+\pi 2Circ)}{w2-m\,w2-wBar-w2\,\pi 2Circ+m\,w2\,\pi 2Circ}\right\}\right\}$$

```
πAvR0 := qEqB π1'[0] + (1 - qEqB) π2'[0]
```

```
πAvR0Term1 = πAvR0 /. πiDR0Rule[[1]]
```

$$-\frac{(-1+m)(-1+qEqB)\,qEqB\,w14\,(-1+\pi 1Circ)(\pi 1Circ-\pi 2Circ)}{w1-m\,w1-wBar-w1\,\pi 1Circ+m\,w1\,\pi 1Circ} + \frac{(-1+m)(1-qEqB)\,qEqB\,w14\,(-1+\pi 2Circ)(-\pi 1Circ+\pi 2Circ)}{w2-m\,w2-wBar-w2\,\pi 2Circ+m\,w2\,\pi 2Circ}$$

$$\pi\text{AvR0Term2} = \frac{(1-m)\ (1-\text{qEqB})\ \text{qEqB}\ \text{w14}\ (1-\pi\text{1Circ})\ (\pi\text{2Circ}-\pi\text{1Circ})}{\text{wBar}-(1-m)\ (1-\pi\text{1Circ})\ \text{w1}} + -\frac{(1-m)\ (1-\text{qEqB})\ \text{qEqB}\ \text{w14}\ (1-\pi\text{2Circ})\ (\pi\text{2Circ}-\pi\text{1Circ})}{\text{wBar}-(1-m)\ (1-\pi\text{2Circ})\ \text{w2}};$$

```
πAvR0Term1 - πAvR0Term2 // FullSimplify
```

0

$$\pi\text{AvR0Term3} = (1-m)\ \text{qEqB}\ (1-\text{qEqB})\ \text{w14}\ (\pi\text{2Circ}-\pi\text{1Circ})\ \frac{1}{\text{wBar}}\left(\frac{1-\pi\text{1Circ}}{1-(1-m)\ (1-\pi\text{1Circ})\ \frac{\text{w1}}{\text{wBar}}}-\frac{1-\pi\text{2Circ}}{1-(1-m)\ (1-\pi\text{2Circ})\ \frac{\text{w2}}{\text{wBar}}}\right);$$

```
πAvR0Term1 - πAvR0Term3 // FullSimplify
```

0

```
eq1LHS[π1[r], π2[r]] == eq1RHS[π1[r], π2[r]] /. λRules /. r → 0
```

$$\text{Log}[1-\pi1[0]] == -\frac{(1-m)\ \text{w1}\ \pi1[0]}{\text{wBar}}$$

```
eq2LHS[π1[r], π2[r]] == eq2RHS[π1[r], π2[r]] /. λRules /. r → 0
```

$$\text{Log}[1-\pi2[0]] == -\frac{(1-m)\ \text{w2}\ \pi2[0]}{\text{wBar}}$$

- Implementation 2: in terms of the extinction probabilities Q_i

```
eq1LHSalt[Q1_, Q2_] := Log[Q1]
eq1RHSalt[Q1_, Q2_] := -λ11 (1 - Q1) - λ12 (1 - Q2)
eq2LHSalt[Q1_, Q2_] := Log[Q2]
eq2RHSalt[Q1_, Q2_] := -λ21 (1 - Q1) - λ22 (1 - Q2)
```

```
eq1LHSalt[Q1[r], Q2[r]] == eq1RHSalt[Q1[r], Q2[r]] /. λRules
```

$$\text{Log}[\text{Q1}[r]] == -\frac{(1-m)\ (\text{w1}-(1-\text{qEqB})\ r\ \text{w14})\ (1-\text{Q1}[r])}{\text{wBar}}-\frac{(1-m)\ (1-\text{qEqB})\ r\ \text{w14}\ (1-\text{Q2}[r])}{\text{wBar}}$$

```
D[eq1LHSalt[Q1[r], Q2[r]] /. λRules, r] == D[eq1RHSalt[Q1[r], Q2[r]] /. λRules, r]
```

$$\frac{\text{Q1}'[r]}{\text{Q1}[r]} == \frac{(1-m)\ (1-\text{qEqB})\ \text{w14}\ (1-\text{Q1}[r])}{\text{wBar}}-\frac{(1-m)\ (1-\text{qEqB})\ \text{w14}\ (1-\text{Q2}[r])}{\text{wBar}}+\frac{(1-m)\ (\text{w1}-(1-\text{qEqB})\ r\ \text{w14})\ \text{Q1}'[r]}{\text{wBar}}+\frac{(1-m)\ (1-\text{qEqB})\ r\ \text{w14}\ \text{Q2}'[r]}{\text{wBar}}$$

```
D[eq2LHSalt[Q1[r], Q2[r]] /. λRules, r] == D[eq2RHSalt[Q1[r], Q2[r]] /. λRules, r]
```

$$\frac{\text{Q2}'[r]}{\text{Q2}[r]} == -\frac{(1-m)\ \text{qEqB}\ \text{w14}\ (1-\text{Q1}[r])}{\text{wBar}}+\frac{(1-m)\ \text{qEqB}\ \text{w14}\ (1-\text{Q2}[r])}{\text{wBar}}+\frac{(1-m)\ \text{qEqB}\ r\ \text{w14}\ \text{Q1}'[r]}{\text{wBar}}+\frac{(1-m)\ (-\text{qEqB}\ r\ \text{w14}+\text{w2})\ \text{Q2}'[r]}{\text{wBar}}$$

```
Deq1alt = FullSimplify[
  D[eq1LHSalt[Q1[r], Q2[r]] /. λRules, r] == D[eq1RHSalt[Q1[r], Q2[r]] /. λRules, r]]
```

$$\frac{1}{\text{wBar}\ \text{Q1}[r]}(\text{wBar}\ \text{Q1}'[r]+(-1+m)\ \text{Q1}[r]\ (\text{w1}\ \text{Q1}'[r]+(-1+\text{qEqB})\ \text{w14}\ (\text{Q1}[r]-\text{Q2}[r]+r\ (\text{Q1}'[r]-\text{Q2}'[r])))) == 0$$

```
Deq2alt = FullSimplify[
  D[eq2LHSalt[Q1[r], Q2[r]] /. λRules, r] == D[eq2RHSalt[Q1[r], Q2[r]] /. λRules, r]]
```

$$\frac{\text{Q2}'[r]}{\text{Q2}[r]}+\frac{1}{\text{wBar}}(-1+m)\ (\text{qEqB}\ \text{w14}\ (\text{Q1}[r]-\text{Q2}[r]+r\ \text{Q1}'[r])+(-\text{qEqB}\ r\ \text{w14}+\text{w2})\ \text{Q2}'[r]) == 0$$

```
Deq1alt /. {r → 0}
```

$$\frac{1}{\text{wBar Q1}[0]}\left(\text{wBar Q1}'[0]+(-1+m)\,\text{Q1}[0]\left((-1+\text{qEqB})\,\text{w14}\,(\text{Q1}[0]-\text{Q2}[0])+\text{w1 Q1}'[0]\right)\right)=0$$

```
Deq2alt /. {r → 0}
```

$$\frac{\text{Q2}'[0]}{\text{Q2}[0]}+\frac{(-1+m)\,(\text{qEqB w14}\,(\text{Q1}[0]-\text{Q2}[0])+\text{w2 Q2}'[0])}{\text{wBar}}=0$$

```
Deq1R0alt = Deq1alt /. {r → 0} /. {Q1[0] → Q1Circ, Q2[0] → Q2Circ}
```

$$\frac{1}{\text{Q1Circ wBar}}\left(\text{wBar Q1}'[0]+(-1+m)\,\text{Q1Circ}\left((\text{Q1Circ}-\text{Q2Circ})\,(-1+\text{qEqB})\,\text{w14}+\text{w1 Q1}'[0]\right)\right)=0$$

```
Deq2R0alt = Deq2alt /. {r → 0} /. {Q1[0] → Q1Circ, Q2[0] → Q2Circ}
```

$$\frac{\text{Q2}'[0]}{\text{Q2Circ}}+\frac{(-1+m)\,((\text{Q1Circ}-\text{Q2Circ})\,\text{qEqB w14}+\text{w2 Q2}'[0])}{\text{wBar}}=0$$

Now we solve for $Q_1'[0]$ and $Q_2'[0]$:

```
QiDR0Rule = Solve[{Deq1R0alt, Deq2R0alt}, {Q1′[0], Q2′[0]}]
```

$$\left\{\left\{\text{Q1}'[0]\to-\frac{(-1+m)\,\text{Q1Circ}\,(\text{Q1Circ}-\text{Q2Circ})\,(-1+\text{qEqB})\,\text{w14}}{-\text{Q1Circ w1}+m\,\text{Q1Circ w1}+\text{wBar}},\right.\right.$$
$$\left.\left.\text{Q2}'[0]\to-\frac{(-1+m)\,(\text{Q1Circ}-\text{Q2Circ})\,\text{Q2Circ qEqB w14}}{-\text{Q2Circ w2}+m\,\text{Q2Circ w2}+\text{wBar}}\right\}\right\}$$

```
QAvR0 := qEqB Q1′[0] + (1 - qEqB) Q2′[0]
QAvR0Term1 = QAvR0 /. QiDR0Rule⟦1⟧
```

$$-\frac{(-1+m)\,\text{Q1Circ}\,(\text{Q1Circ}-\text{Q2Circ})\,(-1+\text{qEqB})\,\text{qEqB w14}}{-\text{Q1Circ w1}+m\,\text{Q1Circ w1}+\text{wBar}}-\frac{(-1+m)\,(\text{Q1Circ}-\text{Q2Circ})\,\text{Q2Circ}\,(1-\text{qEqB})\,\text{qEqB w14}}{-\text{Q2Circ w2}+m\,\text{Q2Circ w2}+\text{wBar}}$$

$$\text{QAvR0Term2}=\frac{(1-m)\,(1-\text{qEqB})\,\text{qEqB Q1Circ}\,(\text{Q2Circ}-\text{Q1Circ})\,\text{w14}}{\text{wBar}-(1-m)\,\text{Q1Circ w1}}-\frac{(1-m)\,(1-\text{qEqB})\,\text{qEqB}\,(\text{Q2Circ}-\text{Q1Circ})\,\text{Q2Circ w14}}{\text{wBar}-(1-m)\,\text{Q2Circ w2}};$$

```
QAvR0Term1 - QAvR0Term2 // FullSimplify
```

0

$$\text{QAvR0Term3}=(1-m)\,(1-\text{qEqB})\,\text{qEqB w14}\,(\text{Q2Circ}-\text{Q1Circ})\,\frac{1}{\text{wBar}}\left(\frac{\text{Q1Circ}}{1-(1-m)\,\text{Q1Circ}\,\frac{\text{w1}}{\text{wBar}}}-\frac{\text{Q2Circ}}{1-(1-m)\,\text{Q2Circ}\,\frac{\text{w2}}{\text{wBar}}}\right);$$

```
QAvR0Term1 - QAvR0Term3 // FullSimplify
```

0

$$\text{QAvR0Term4}=(1-m)\,(1-\text{qEqB})\,\text{qEqB w14}\,(-\pi\text{2Circ}+\pi\text{1Circ})\,\frac{1}{\text{wBar}}\left(\frac{1-\pi\text{1Circ}}{1-(1-m)\,(1-\pi\text{1Circ})\,\frac{\text{w1}}{\text{wBar}}}-\frac{1-\pi\text{2Circ}}{1-(1-m)\,(1-\pi\text{2Circ})\,\frac{\text{w2}}{\text{wBar}}}\right);$$

```
QAvR0Term4 - QAvR0Term3 /. {π1Circ → 1 - s1Circ, π2Circ → 1 - s2Circ}
```

$$-\frac{1}{\text{wBar}}(1-\text{m})(-\text{Q1Circ}+\text{Q2Circ})(1-\text{qEqB})\,\text{qEqB}\,\text{w14}\left(\frac{\text{Q1Circ}}{1-\frac{(1-\text{m})\,\text{Q1Circ}\,\text{w1}}{\text{wBar}}}-\frac{\text{Q2Circ}}{1-\frac{(1-\text{m})\,\text{Q2Circ}\,\text{w2}}{\text{wBar}}}\right)+$$
$$\frac{1}{\text{wBar}}(1-\text{m})(1-\text{qEqB})\,\text{qEqB}\,(-\text{s1Circ}+\text{s2Circ})\,\text{w14}\left(\frac{\text{s1Circ}}{1-\frac{(1-\text{m})\,\text{s1Circ}\,\text{w1}}{\text{wBar}}}-\frac{\text{s2Circ}}{1-\frac{(1-\text{m})\,\text{s2Circ}\,\text{w2}}{\text{wBar}}}\right)$$

```
QAvR0Term4
```

$$\frac{1}{\text{wBar}}(1-\text{m})(1-\text{qEqB})\,\text{qEqB}\,\text{w14}\left(\frac{1-\pi\text{1Circ}}{1-\frac{(1-\text{m})\,\text{w1}\,(1-\pi\text{1Circ})}{\text{wBar}}}-\frac{1-\pi\text{2Circ}}{1-\frac{(1-\text{m})\,\text{w2}\,(1-\pi\text{2Circ})}{\text{wBar}}}\right)(\pi\text{1Circ}-\pi\text{2Circ})$$

```
πAvR0Term3
```

$$\frac{1}{\text{wBar}}(1-\text{m})(1-\text{qEqB})\,\text{qEqB}\,\text{w14}\left(\frac{1-\pi\text{1Circ}}{1-\frac{(1-\text{m})\,\text{w1}\,(1-\pi\text{1Circ})}{\text{wBar}}}-\frac{1-\pi\text{2Circ}}{1-\frac{(1-\text{m})\,\text{w2}\,(1-\pi\text{2Circ})}{\text{wBar}}}\right)(-\pi\text{1Circ}+\pi\text{2Circ})$$

We expcect the derivation of $\overline{s}(r)$ at $r = 0$ to be minus the derivation of $\overline{\pi}(r)$ at $r = 0$:

```
(-QAvR0Term4) - πAvR0Term3 // Simplify
```

0

This seems to be fine.

Summary

In summary, we have found the derivation of $\overline{\pi}(r)$ with respect to r, evalutated at $r = 0$, to be

$$\overline{\pi}'(0) = \frac{d}{dr}[\hat{q}_B\,\pi_1(r) + (1-\hat{q}_B)\,\pi_2(r)]\,|_{r=0} = (1-m)\,\hat{q}_B(1-\hat{q}_B)\left(\pi_2^\circ-\pi_1^\circ\right)\frac{w_{14}}{\overline{w}}\left(\frac{1-\pi_1^\circ}{1-(1-m)\,(1-\pi_1^\circ)\,w_1/\overline{w}}-\frac{1-\pi_2^\circ}{1-(1-m)\,(1-\pi_2^\circ)\,w_2/\overline{w}}\right). \quad (9)$$

We note that if we set $m = 0$ and take $\hat{q}_B$ as the equilibrium frequency of allele B_1 if the polymorphism at locus $\mathcal{B}$ is maintained by heterozygote superiority, then Eq. (9) is identical to Eq. (32) of Ewens (1967), except that Ewens called w_4 and w_{22} what we call w_2 and w_{14}, respectively.

Setting $r = 0$ in Eq. (6), we obtain

```
eq1R0 := eq1LHSalt[Q1[r], Q2[r]] == eq1RHSalt[Q1[r], Q2[r]] /. λRules /. r → 0
eq2R0 := eq2LHSalt[Q1[r], Q2[r]] == eq2RHSalt[Q1[r], Q2[r]] /. λRules /. r → 0
{eq1R0, eq2R0} // TableForm
```

$$\text{Log}[\text{Q1}[0]] == -\frac{(1-\text{m})\,\text{w1}\,(1-\text{Q1}[0])}{\text{wBar}}$$
$$\text{Log}[\text{Q2}[0]] == -\frac{(1-\text{m})\,\text{w2}\,(1-\text{Q2}[0])}{\text{wBar}}$$

or, replacing $Q_i(0) \to 1-\pi_i^\circ$,

$$(1-m)\,\frac{w_1}{\overline{w}} = -\left(\pi_1^\circ\right)^{-1}\,\text{Log}\left(1-\pi_1^\circ\right) \quad (10)$$

and

$$(1-m)\,\frac{w_2}{\overline{w}} = -\left(\pi_2^\circ\right)^{-1}\,\text{Log}\left(1-\pi_2^\circ\right) \quad (11)$$

Plugging Eqs. (10) and (11) into Eq. (9), we obtain

$$\overline{\pi}'(0) = (1-m)\,\hat{q}_B(1-\hat{q}_B)\left(\pi_2^\circ-\pi_1^\circ\right)\frac{w_{14}}{\overline{w}}\left(\frac{1-\pi_1^\circ}{1+(\pi_1^\circ)^{-1}\,(1-\pi_1^\circ)\,\text{Log}(1-\pi_1^\circ)}-\frac{1-\pi_2^\circ}{1+(\pi_2^\circ)^{-1}\,(1-\pi_2^\circ)\,\text{Log}(1-\pi_2^\circ)}\right). \quad (12)$$

Let us assume that $\pi_1^\circ > \pi_2^\circ$. Further, we note that the function $\frac{1-x}{1+\frac{1}{x}(1-x)\,\mathrm{Log}(1-x)}$ is a decreasing function of x, because its numerator is a decreasing function and its denominator an increasing function of x. Taken together, this suggest that the derivative in Eq. (12) is alsways positive and hence the optimal recombination rate is always larger than zero.

```
Plot[(1 - x) / (1 + 1/x (1 - x) Log[1 - x]), {x, 0, 1}, PlotRange → {{0, 1}, Automatic}]
```



However, we now recall that our general assumption was that $w_2 < \overline{w}$ and, with complete linkage ($r = 0$), the mutant A_1 cannot invade if it initially occurs on the B_2 background. We therefore set $\pi_2^\circ = 0$ in Eq. (9), which yields

$$\overline{\pi}'(0) = (1-m)\,\hat{q}_B(1-\hat{q}_B)\,\pi_1^\circ\,\frac{w_{14}}{\overline{w}}\left(\frac{\overline{w}}{\overline{w}-(1-m)\,w_2} - \frac{1-\pi_1^\circ}{1-(1-m)(1-\pi_1^\circ)\,w_1/\overline{w}}\right). \quad (13)$$

This is poistive, if

```
FullSimplify[Reduce[wBar / (wBar - (1 - m) w2) - (1 - π1Star) / (1 - (1 - m) (1 - π1Star) w1 / wBar) > 0 /. {m → 0}, w2],
 Assumptions → {0 < m < 1, 0 < π1Star < 1, 0 < w2 < wBar < w1}]
wBar + w1 π1Star < w1 || w1 < w2 + (w1 - w2 + wBar) π1Star
w2 (1 - π1Star) > w1 (1 - π1Star) - wBar π1Star
```

from which it follows that

```
w2 > w1 - π1Star / (1 - π1Star) wBar
FullSimplify[Solve[wBar / (wBar - (1 - m) w2) - (1 - π1Star) / (1 - (1 - m) (1 - π1Star) w1 / wBar) == 0 /. {m → 0}, w2]]
{{w2 → w1 + (wBar π1Star) / (-1 + π1Star)}}
```

if there is no migration ($m = 0$) and, more generally

```
FullSimplify[Solve[wBar / (wBar - (1 - m) w2) - (1 - π1Star) / (1 - (1 - m) (1 - π1Star) w1 / wBar) == 0, w2]]
{{w2 → ((-1 + m) w1 (-1 + π1Star) - wBar π1Star) / ((-1 + m) (-1 + π1Star))}}
((-1 + m) w1 (-1 + π1Star) - wBar π1Star) / ((-1 + m) (-1 + π1Star)) - (w1 - (wBar π1Star) / ((1 - m) (1 - π1Star))) // Simplify
0
```

$$w_2 < w_1 - \overline{w}\,\frac{\pi_1^\circ}{(1-m)\,(1-\pi_1^\circ)} \Leftrightarrow w_1 - w_2 > \overline{w}\,\frac{\pi_1^\circ}{(1-m)\,(1-\pi_1^\circ)}. \tag{14}$$

Again, setting $m = 0$, we obtain the condition found earlier by Ewens (his Eq. (36)). Although condition (14) is fully generic, it is not very informative, because $\overline{w}$ depends on the equilibrium frequency $\hat{q}_B$ and because we do not know π_1° explicitly. Therefore, in the following, we assume additive fitnesses and attempt to obtain approximate explicit conditions. We first consider the case of a monomorphic continent ($q_c = 0$) and will then turn to the polymorphic continent ($0 < q_c < 1$).

Monomorphic continent with additive fitnesses

Description

We start with the system of equations that must be solved to find the invasion probabilities $\pi_i = 1 - s_i$, where s_i is the probability of extinction of type i. We distinguish between the extinction probability s_i and the argument s_i of the probability-generating function (pgf). Throughout, we assume additive fitnesses and a monomorphic continent ($q_c = 0$). The s_i are obtained as the smallest solutions s_i between 0 and 1 to the following system of equations:

$$f_1\,(s_1,\ s_2) = e^{((b\,(1+b+a\,m)-m\,(1-a+b)\,r)\,s_1+(m\,(1-a+b)\,r)\,s_2-b\,(1+b+a\,m))/(b\,(1-a+b))} = s_1 \tag{1}$$

$$f_2\,(s_1,\ s_2) = e^{(((b-(1-a)\,m)\,r)\,s_1+(b\,(1+m\,(a-b))-(b-(1-a)\,m)\,r)\,s_2-b\,(1+\,m\,(a-b)))/(b\,(1-a+b))} = s_2 \tag{2}$$

where the $f_i(s_1,\ s_2)$ are the pgfs of the offspring distribution.

With complete linkage ($r = 0$), these two equations decouple and become

$$f_1\,(s_1,\ s_2) = f_1(s_1) = e^{-\frac{1+b+a\,m}{1-a+b}\,(1-s_1)} = s_1 \tag{3}$$

$$f_2\,(s_1,\ s_2) = f_2(s_2) = e^{-\frac{1+a\,m-b\,m}{1-a+b}\,(1-s_2)} = s_2. \tag{4}$$

Both equations are of the form

$$e^{-Z_i(1-s_i)} = s_i \Leftrightarrow e^{-Z_i\,x_i} = 1 - x_i. \tag{5}$$

Equation (5) has a solution $s_i \in [0,\ 1) \Leftrightarrow Z_i > 1$. Otherwise ($Z_i \leq 1$), it has only one solution, $s_i = 1$. This is easily verified by considering the plots below, and by realising that $e^{-Z_i\,x_i} = 1 - x_i$ always has a solution at $x_{i(1)} = 0$, but it only has a second solution $x_{i(2)} \in (0,\ 1]$ if $Z_i > 1$. Note that $x_{i(1)} = 0 < x_{i(2)}$, and therefore $s_{i(2)} = 1 - x_{i(2)} < s_{i(1)} = 1$. The extinction probability is given by the smallest solution s_i between 0 and 1. Hence, we have $0 \leq s_i < 1$ if and only if $Z_i > 1$.

```
Manipulate[
 Plot[{1 - x, Exp[-Z x]}, {x, 0, 10}, PlotStyle → {{Black, Dashed}, {Black, Thick}},
  PlotRange → {{-0.2, 1.2}, {-0.2, 1.2}}, AxesLabel → {x, e^(-Z x)}], {{Z, 1}, -4, 4}]
```




```
eqf1 = s1 == Exp[(EE + FF r) s1 - FF r s2 - EE];
eqf2 = s2 == Exp[HH r s1 + (JJ + HH r) s2 - JJ];
{eqf1, eqf2} /. r → 0 // Simplify // TableForm
```

$e^{EE\,(-1+s1)} == s1$
$e^{JJ\,(-1+s2)} == s2$

Implementation

■ Exact probability-generating functions (pgf)

```
Clear[pgf1, pgf2]
```

We start with Equations (1) and (2) if full form.

```
pgf1[s1_, s2_] := e^((b (1+b+a m)-m (1-a+b) r) s1+(m (1-a+b) r) s2-b (1+b+a m))/(b (1-a+b))
pgf2[s1_, s2_] := e^(((b-(1-a) m) r) s1+(b (1+m (a-b))-(b-(1-a) m) r) s2-b (1+ m (a-b)))/(b (1-a+b))
```

As a check, we assess if they agree with an alternative way of writing them (see *Mathematica* Notebook '2LocContIsland_Stoch_Discr.nb', expressions 'pgf1Add' and 'pgf2Add').

$$\text{pgf1[s1, s2]} - e^{\frac{b^2\,(-1+s1)+(-1+a)\,m\,r\,(s1-s2)+b\,(-1+a\,m\,(-1+s1)+s1-m\,r\,s1+m\,r\,s2)}{b\,(1-a+b)}} \text{ // Simplify}$$

0

$$\text{pgf2[s1, s2]} - e^{\frac{(-1+a)\,m\,r\,(s1-s2)+b^2\,(m-m\,s2)+b\,(-1+r\,s1+a\,m\,(-1+s2)+s2-r\,s2)}{b\,(1-a+b)}} \text{ // Simplify}$$

0

■ Approximate pgfs for $r = 0$

```
pgf1NoRec[s1_, s2_] := pgf1[s1, s2] /. {r → 0} // Simplify
pgf2NoRec[s1_, s2_] := pgf2[s1, s2] /. {r → 0} // Simplify
```

As usual, let $\{Q_1, Q_2\}$ be the smallest solution between 0 and 1 to the system $f_i(s_1, s_2) = s_i$ $\{i = 1, 2\}$.

```
pgf1NoRec[Q1, Q2]
```

$$e^{-\frac{(1+b+a\,m)\,(-1+Q1)}{-1+a-b}}$$

```
pgf2NoRec[Q1, Q2]
```

$e^{-\frac{(1+a m-b m)(-1+Q2)}{-1+a-b}}$

As mentioned above, for $r = 0$, the two equations decouple (the first is independent of Q_2 and the second is independent of Q_1).

We define

```
Z1 := (1 + b + a m) / (1 - a + b);
Z2 := (1 + m (a - b)) / (1 - a + b);
FullSimplify[Reduce[Z2 < 1], Assumptions → {0 < a < b < 1, 0 < m < 1}]
```

```
True
```

```
FullSimplify[Reduce[Z1 < 1], Assumptions → {0 < a < b < 1, 0 < m < 1}]
```

```
False
```

where we note already here that $Z_2 < 1$ and $Z_1 > 1$ always, given our assumption of $a < b$ and given the natural restriction of $0 < m < 1$.

We reformulate the pgfs for $r = 0$ in terms of Z_i in a generic form.

```
pgfNR[s_] := Exp[-Z (1 - s)]
```

```
Solve[Q == pgfNR[Q] /. Z → Z1, Q]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{Q \to \frac{(-1+a-b)\,\mathrm{ProductLog}\left[\frac{e^{\frac{1}{-1+a-b}+\frac{b}{-1+a-b}+\frac{a m}{-1+a-b}}(1+b+a m)}{-1+a-b}\right]}{1+b+a m}\right\}\right\}$$

An implicit solution is still not available in general, because the equation $y = e^{-Z_i y}$ is transcendental.

Approximate solution for $r = 0$

- Assuming small evolutionary forces – Haldane's (1927) approximation

For the time being, we omit the subscripts i. Recall that $Q \in [0, 1)$ if and only if $Z > 1$. An interesting case is when invasion is just possible, *i.e.* when $\pi = 1 - Q$ is close to zero. This is equivalent to Z being close to but larger than 1. We may therefore use the Ansatz

$$Z = 1 + \epsilon, \tag{6}$$

where $\epsilon > 0$ is small.

```
ruleZ := {Z → 1 + ϵ}
```

Of course, just by making this substitution, the equation of interest remains transcendental and, generically, it is not possible to find an explicit solution.

```
pgfNR[Q]
```

$e^{-(1-Q) Z}$

```
Series[pgfNR[Q] /. {Z → (1 + ϵ)}, {ϵ, 0, 1}] // Normal
```

$e^{-1+Q} + e^{-1+Q}(-1+Q)\,\epsilon$

```
Solve[e^(-1+Q) + e^(-1+Q) (-1 + Q) ϵ == Q, Q]
```

Solve::nsmet : This system cannot be solved with the methods available to Solve. ≫

$\mathrm{Solve}\left[e^{-1+Q} + e^{-1+Q}(-1+Q)\,\epsilon == Q,\ Q\right]$

However, with $Z = 1 + \epsilon$, we know that $Q = Q(\epsilon)$ is a function of ϵ. Specifically, we have

$$Q(\epsilon) = e^{-(1+\epsilon)(1-Q)} \tag{7}$$

```
pgfNR[s] /. ruleZ
```

$e^{-(1-s)(1+\epsilon)}$

A Taylor series expansion of $Q(\epsilon)$ around $\epsilon = 0$ yields

```
Series[e^(-(1-Q) (1+ϵ)), {ϵ, 0, 2}] // Normal
```

$e^{-1+Q} + e^{-1+Q}(-1+Q)\,\epsilon + \frac{1}{2} e^{-1+Q}(-1+Q)^2\,\epsilon^2$

We know that, for $\epsilon > 0$ small, Q must be close to 1. So, we expand $e^{-(1-Q)(1+\epsilon)}$ around $Q = Q_0 = 0$.

```
term7 = Series[e^(-(1+ϵ) (1-Q)), {Q, 1, 2}] // Simplify // Normal
```

$1 + (-1+Q)(1+\epsilon) + \frac{1}{2}(-1+Q)^2(1+\epsilon)^2$

```
testTerm7 = 1 - (1 - Q) (1 + ϵ) + 1/2 (1 - Q)^2 (1 + ϵ)^2
```

$1 - (1-Q)(1+\epsilon) + \frac{1}{2}(1-Q)^2(1+\epsilon)^2$

```
testTerm7 - term7 // Simplify
```

0

Notice that it is important to expand this up to order $O[Q]^2$. Equating this and solving for Q, we obtain

```
Solve[term7 == Q, Q]
```

$\left\{\{Q \to 1\}, \left\{Q \to \frac{1+\epsilon^2}{(1+\epsilon)^2}\right\}\right\}$

Approximating the second solution ($Q \neq 1$) assuming $\epsilon > 0$ small, we find

```
Series[(1 + ϵ^2)/(1 + ϵ)^2, {ϵ, 0, 1}]
```

$1 - 2\,\epsilon + O[\epsilon]^2$

$$Q = 1 - \pi \approx 1 - 2\,\epsilon, \quad (8)$$

and hence

$$\pi \approx 2\,\epsilon \quad (9)$$

ignoring terms of order ϵ^2 and higher.

To identify ϵ in the cases where $Z = Z_1$ or $Z = Z_2$, we write Z_i in the form $Z_i = 1 + \epsilon_i$.

```
Z1
```

$\frac{1 + b + a\,m}{1 - a + b}$

```
Z2
```

$\frac{1 + (a-b)\,m}{1 - a + b}$

Assuming small evolutionary forces, *i.e.* letting $a \to \alpha\,\epsilon_i$, $b \to \beta\,\epsilon_i$, $m \to \mu\,\epsilon_i$ with $\epsilon_i > 0$ small, we expand $Z_i = Z_i(\epsilon_i)$ around $\epsilon_i = 0$:

```
Z1 /. ruleSmallForces /. {ϵ → ϵ1}
```

$\frac{1 + \beta\,\epsilon 1 + \alpha\,\epsilon 1^2\,\mu}{1 - \alpha\,\epsilon 1 + \beta\,\epsilon 1}$

```
Z2 /. ruleSmallForces /. {ϵ → ϵ2}
```

$\frac{1 + \epsilon 2\,(\alpha\,\epsilon 2 - \beta\,\epsilon 2)\,\mu}{1 - \alpha\,\epsilon 2 + \beta\,\epsilon 2}$

```
Series[Z1 /. ruleSmallForces /. {ϵ → ϵ1}, {ϵ1, 0, 1}]
```

$1 + \alpha\,\epsilon 1 + O[\epsilon 1]^2$

```
Series[Z2 /. ruleSmallForces /. {ϵ → ϵ2}, {ϵ2, 0, 1}]
```

```
1 + (α - β) ϵ2 + O[ϵ2]^2
```

For $\alpha < \beta$, we find that $Z_2 < 1$ always and, therefore, $\pi_2 = 1 - Q_2 = 0$ always. The interesting case is for Z_1, which upon resubstitution of $\alpha \to a/\epsilon_1$becomes

$$Z_1 = 1 + a \tag{10}$$

```
1 + α ϵ1 /. ruleReturnOrigin /. {ϵ → ϵ1}
```

```
1 + a
```

Comparing to our Ansatz in Eq. (6), we identify $\epsilon_1 = a$. Plugging Eq. (10) into Eq. (8), we obtain

$$Q_1 = 1 - \pi_1 \approx 1 - 2\,a. \tag{11}$$

From this, we conclude that

$$\pi_1 = 1 - Q_1 \approx 2\,a, \tag{12}$$

which corresponds to Haldane's (1927) approximation, where a is the advantage of a heterozygote.

To summarise, for $r = 0$ we obtain the approximate conditional invasion probabilities $\{\pi_1, \pi_2\} \approx \{2\,a, 0\}$.

However, $\epsilon_1 = a$ is a very rough approximation; most importantly it is independent of m, which, for a model with migration, is not desirable. In the following, we try an alternative route by assuming only that a is small, but making no further assumption about b and m.

- Assuming small a but arbitrary b and m

Starting from

$$Z_1 = \frac{1 + b + a\,m}{1 - a + b} = 1 + \epsilon_1, \tag{13}$$

we again want to identify ϵ_1. We may write Z_1 as

$$Z_1 = \frac{1 + b + a\,m}{1 - a + b} = \frac{1 + b + a\,m}{1 + b - a} = \frac{1 + \frac{a\,m}{1+b}}{1 - \frac{a}{1+b}} = \left(1 + \frac{a\,m}{1 + b}\right)\frac{1}{1 - \frac{a}{1+b}}. \tag{14}$$

Recalling the geometric series $\frac{1}{1-r} = 1 + r + r^2 + r^3 + ...$, with $r = \frac{a}{1+b}$ in our case, we find

$$Z_1 = \left(1 + \frac{a\,m}{1 + b}\right)\left(1 + \frac{a}{1 + b} + \frac{a^2}{(1 + b)^2} + ...\right), \tag{15}$$

$$\left(1 + \frac{a\,m}{1 + b}\right)\left(1 + \frac{a}{1 + b} + \frac{a^2}{(1 + b)^2}\right) \text{ // Expand}$$

$$1 + \frac{a^2}{(1 + b)^2} + \frac{a}{1 + b} + \frac{a^3\,m}{(1 + b)^3} + \frac{a^2\,m}{(1 + b)^2} + \frac{a\,m}{1 + b}$$

which simplifies to

$$Z_1 \approx 1 + \frac{(m + 1)\,a}{1 + b} + \frac{a^2\,m}{(1 + b)^2}, \tag{16}$$

ignoring terms $O[a]^3$ and higher, or to

$$Z_1 \approx 1 + \frac{(m + 1)\,a}{1 + b}, \tag{17}$$

if we only consider terms up to order $O[a]$. We have thus identified ϵ_1 as

$$\epsilon_1 \approx \frac{(m + 1)\,a}{1 + b}. \tag{18}$$

More directly, we have

```
Z1
```

$$\frac{1+b+a\,m}{1-a+b}$$

```
rule∈1 = FullSimplify[Solve[1 + ∈1 == Z1, ∈1]]
```

$$\left\{\left\{\epsilon 1 \to \frac{a\,(1+m)}{1-a+b}\right\}\right\}$$

```
FullSimplify[Series[∈1 /. rule∈1[[1]] /. a → α ∈, {∈, 0, 1}] /. {α → a / ∈}] // Normal
```

$$\frac{a\,(1+m)}{1+b}$$

```
Series[Z1, {a, 0, 2}] // Normal
```

$$1+\frac{a^2\,(1+m)}{(1+b)^2}+\frac{a\,(1+m)}{1+b}$$

Insertion of ϵ_1 from Eq. (18) into Eq. (8) yields

$$Q_1 = 1-\pi_1 \approx 1-2\,\frac{a\,(m+1)}{b+1} \tag{19}$$

as an approximation to the extinction probability Q_1. Recall that we set $r = 0$. Recall further that $s_2 = 1$ whenever $a < b$, so that $Q_2 = 1-\pi_2 = 1$ always in our case.

Returning to the more accurate approximation of ϵ_1 in Eq. (16), we find

$$Q_1 = 1-\pi_1 \approx 1-2\left[\frac{(m+1)\,a}{1+b}+\frac{a^2\,m}{(1+b)^2}\right] = 1-\frac{2\,a\,[1+b+(1+a+b)\,m]}{(1+b)^2}. \tag{20}$$

The invasion probability of interest at $r = 0$ is $\pi_1 \approx \frac{2\,a\,[1+b+(1+a+b)\,m]}{(1+b)^2}$.

```
π1Approx1 := 2 (1 + m) a / (1 + b);
π1Approx2 := 2 a (1 + b + (1 + a + b) m) / (1 + b)^2;
Q1Approx1 := 1 - π1Approx1;
Q1Approx2 := 1 - π1Approx2;
Q2Approx1 := 1;
Q2Approx2 = Q2Approx1;
```

The steps in the preceding two sub-sections have been suggested by Josef Hofbauer (personal communication, March 2012).

Derivatives of $f_i(s_1, s_2)$ for $r > 0$ but small

Taking the logarithm on both sides, Eqs. (1) and (2) can be written alternatively as

$$(\mathcal{A}+\mathcal{B}\,r)\,s_1+\mathcal{C}\,r\,s_2-\mathcal{D} = \log s_1 \tag{21}$$

$$\mathcal{E}\,r\,s_1+(\mathcal{F}+\mathcal{G}\,r)\,s_2-\mathcal{H} = \log s_2, \tag{22}$$

where

```
ruleA := {A → (1 + b + a m)/(1 - a + b)} (* = -D *) (* Called E in the manuscript *)
ruleB := {B → -m/b} (* = -C *) (* Called F in the manuscript *)
ruleC := {C → m/b} (* = -B *) (* And hence equal to -F,
with F as in the manuscript *)
ruleD := {D → (1 + b + a m)/(1 - a + b)} (* = A *) (* Equal to E in the manuscript *)
ruleE := {E → (b - (1 - a) m)/(b (1 - a + b))} (* = -G *) (* Called H in the manuscript *)
ruleF := {F → (1 + m (a - b))/(1 - a + b)} (* Called J in the manuscript *)
ruleG := {G → -(b - (1 - a) m)/(b (1 - a + b))} (* = -E *)
(* And hence equal to -H in the manuscript *)
ruleH := {H → (1 + m (a - b))/(1 - a + b)} (* = F *) (* And hence equal to J in the manuscript *)
```

We recall that the Q_i ($i = 1, 2$) are the smallest solutions between 0 and 1 to Eqs. (21) and (22). Therefore, the Q_i fulfill

$$(E + F\,r)\,Q_1 - F\,r\,Q_2 - E = \log Q_1 \tag{23}$$

$$H\,r\,Q_1 + (J - H\,r)\,Q_2 - J = \log Q_2 \tag{24}$$

where

```
ruleE := {EE → (1 + b + a m)/(1 - a + b)}
ruleF := {F → -m/b}
ruleH := {H → (b - (1 - a) m)/(b (1 - a + b))}
ruleJ := {J → (1 + m (a - b))/(1 - a + b)}
```

In the following, we use Eqs. (23) and (24).

We note from Eqs. (23) and (24) that the $Q_i = 1 - \pi_i$ are functions of r, *i.e.* $Q_i = Q_i(r)$ with $i \in \{1, 2\}$. This system of equations implicitly defines the Q_i, but – as shown above – an explicit solution cannot be found. Our goal is to find the derivative of $Q_i(r)$ with respect to r, evaluated at $r \to 0$. In the absence of an explicit solution we resort to implicit differentiation.

```
implF1LHS[Q1_, Q2_] := (EE + F r) Q1 - F r Q2 - EE
implF1RHS[Q1_, Q2_] := Log[Q1]
implF2LHS[Q1_, Q2_] := H r Q1 + (J - H r) Q2 - J
implF2RHS[Q1_, Q2_] := Log[Q2]

implF1LHS[Q1[r], Q2[r]]

-EE + (EE + F r) Q1[r] - F r Q2[r]

(* Take Qi=Qi(r) and differentiate with respect to
 r. Do this for the LHS and RHS of both equations above. *)
DImplF1LHS = D[implF1LHS[Q1[r], Q2[r]], r]

F Q1[r] - F Q2[r] + (EE + F r) Q1'[r] - F r Q2'[r]

DImplF1RHS = D[implF1RHS[Q1[r], Q2[r]], r]

Q1'[r]/Q1[r]
```

```
DImplF2LHS = D[implF2LHS[Q1[r], Q2[r]], r]
```

H Q1[r] - H Q2[r] + H r Q1′[r] + (J - H r) Q2′[r]

```
DImplF2RHS = D[implF2RHS[Q1[r], Q2[r]], r]
```

$$\frac{Q2'[r]}{Q2[r]}$$

From this, we obtain two equations for the derivatives $Q_1'(r)$ and $Q_2'(r)$ that also contain $Q_1(r)$ and $Q_2(r)$:

```
Deqn1 = DImplF1LHS == DImplF1RHS
```

$$F\,Q1[r] - F\,Q2[r] + (EE + F\,r)\,Q1'[r] - F\,r\,Q2'[r] == \frac{Q1'[r]}{Q1[r]}$$

```
Deqn2 = DImplF2LHS == DImplF2RHS
```

$$H\,Q1[r] - H\,Q2[r] + H\,r\,Q1'[r] + (J - H\,r)\,Q2'[r] == \frac{Q2'[r]}{Q2[r]}$$

$$F\,Q_1(r) - F\,Q_2(r) + (E + F\,r)\,Q_1'(r) - F\,r\,Q_2'(r) = \frac{Q_1'(r)}{Q_1(r)} \tag{25}$$

$$H\,Q_1(r) - H\,Q_2(r) + H\,r\,Q_1'(r) + (J - H\,r)\,Q_2'(r) = \frac{Q_2'(r)}{Q_2(r)} \tag{26}$$

We want to solve for $Q_1'(r)$ and $Q_2'(r)$ at the position $r = 0$. We plug in our approximations for $Q_1(0)$ given in Eq. (19), and $Q_2(0) = 1$.

```
Solve[{Deqn1, Deqn2} /. r → 0, {Q1′[0], Q2′[0]}] // FullSimplify
```

$$\left\{\left\{Q1'[0] \to \frac{F\,Q1[0]\,(-Q1[0] + Q2[0])}{-1 + EE\,Q1[0]},\ Q2'[0] \to \frac{H\,Q2[0]\,(-Q1[0] + Q2[0])}{-1 + J\,Q2[0]}\right\}\right\}$$

```
solApprox1 = Solve[{Deqn1, Deqn2} /. r → 0 /. {Q1[0] → Q1Approx1, Q2[0] → Q2Approx1},
    {Q1′[0], Q2′[0]}] // Simplify
```

$$\left\{\left\{Q1'[0] \to \frac{2\,a\,F\,(1+m)}{(1+b)\left(EE - \frac{1+b}{1+b-2\,a\,(1+m)}\right)},\ Q2'[0] \to \frac{2\,a\,H\,(1+m)}{(1+b)\,(-1+J)}\right\}\right\}$$

```
solApprox1⟦1⟧⟦All⟧ /.
  {Flatten[{ruleA, ruleB, ruleC, ruleD, ruleE, ruleF, ruleG, ruleH}]} // FullSimplify
```

$$\left\{\left\{Q1'[0] \to \frac{2\,a\,F\,(1+m)}{(1+b)\left(EE - \frac{1+b}{1+b-2\,a\,(1+m)}\right)},\ Q2'[0] \to \frac{2\,a\,H\,(1+m)}{(1+b)\,(-1+J)}\right\}\right\}$$

```
DQ1Approx1 = 2 (1 - a + b) m (1 + b - 2 a (1 + m)) / (b (1 + b) (1 + b + 2 a m));
DQ2Approx1 = 2 a (b - (1 - a) m) / ((a - b) b (1 + b));
```

This yields

$$Q_1'(0) = \frac{2\,(1 - a + b)\,m\,[1 + b - 2\,a\,(1 + m)]}{b\,(1 + b)\,(1 + b + 2\,a\,m)} \tag{27}$$

$$Q_2'(0) = \frac{2\,a\,[b - (1 - a)\,m]}{(a - b)\,b\,(1 + b)}. \tag{28}$$

If the derivative of $Q_i(r)$ at $r = 0$ is positive, the derivative of $\pi_i(r = 0)$ is negative, and vice versa.

```
Simplify[DQ1Approx1 > 0, Assumptions → assumeGeneral]
```

```
1 + b > 2 a (1 + m)
```

```
FullSimplify[DQ2Approx1 > 0, Assumptions → assumeGeneral]
```

```
m > b + a m
```

However, we are interested in the derivative of the weighted average invasion probability, $\bar{\pi}(r)$, at $r = 0$, which we can obtain by noting that

$$\frac{d\bar{\pi}(r)}{r} = \frac{d[\hat{q}\,\pi_1(r) + (1-\hat{q})\,\pi_2(r)]}{dr} = (1-\hat{q})\frac{d\pi_2(r)}{dr} + \hat{q}\frac{d\pi_1(r)}{dr} = (1-\hat{q})\frac{d[1-Q_2(r)]}{dr} + \hat{q}\frac{d[1-Q_1(r)]}{dr} = -\left[\hat{q}\frac{dQ_1(r)}{dr} + (1-\hat{q})\frac{dQ_2(r)}{dr}\right] \quad (29)$$

where $\hat{q} = \frac{b-m(1-a)}{b\,(1+m)}$ is the frequency of the B_1 allele at the initial marginal one-locus migration-selection equilibrium. Setting $r = 0$ and substituting our previous approximations for $Q_i(0)$, we obtain

```
qEqRule := qEq → (b - m (1 - a))/(b (1 + m))
```

```
DπAverApprox1 = -qEq DQ1Approx1 - (1 - qEq) DQ2Approx1 /. qEqRule // FullSimplify
```

```
2 (-1 + a - b) m (b + (-1 + a) m) (2 a^2 + b + b^2 - 2 a (1 + b (2 + m)))
------------------------------------------------------------------------
         b^2 (1 + b) (-a + b) (1 + m) (1 + b + 2 a m)
```

$$\frac{d\bar{\pi}(r)}{r}\Big|_{r=0} = \frac{2\,(1-a+b)\,m\,[b-(1-a)\,m]\,\{2\,a^2+b+b^2-2\,a\,[1+b\,(2+m)]\}}{b^2\,(1+b)\,(a-b)\,(1+m)\,(1+b+2\,a\,m)} \quad (30)$$

When is this positive?

```
DπAverApprox1
```

```
2 (-1 + a - b) m (b + (-1 + a) m) (2 a^2 + b + b^2 - 2 a (1 + b (2 + m)))
------------------------------------------------------------------------
         b^2 (1 + b) (-a + b) (1 + m) (1 + b + 2 a m)
```

```
assumeGeneral
```

```
{0 < a < b < 1, a + b < 1, 0 < r < 0.5, 0 < m < 1}
```

```
condApprox1 = Simplify[DπAverApprox1 > 0, Assumptions → assumeGeneral]
```

```
(b + (-1 + a) m) (2 a^2 + b (1 + b) - 2 a (1 + b (2 + m))) < 0
```

```
condApprox1 // FullSimplify
```

```
(b + (-1 + a) m) (2 a^2 + b + b^2 - 2 a (1 + b (2 + m))) < 0
```

The derivative in Eq. (30) is positive if

$(b + (-1 + a)\,m)\left(2\,a^2 + b + b^2 - 2\,a\,(1 + b\,(2 + m))\right) < 0,$

which can also be written as

```
(b - m (1 - a)) (2 a^2 - 2 a (1 + 2 b) - 2 a b m + b (1 + b))
```

```
(b - (1 - a) m) (2 a^2 + b (1 + b) - 2 a (1 + 2 b) - 2 a b m)
```

```
% - ((b + (-1 + a) m) (2 a^2 + b (1 + b) - 2 a (1 + b (2 + m)))) // FullSimplify
```

```
0
```

$$[b-(1-a)\,m]\left[2\,a^2 + b\,(1+b) - 2\,a\,(1+2\,b) - 2\,a\,b\,m\right] < 0 \quad (31)$$

We want to express this condition in terms of a critical selection coefficient a^*, such that $\frac{d}{dr}\bar{\pi}(r)\,|_{r=0}$is positive whenever $a < a^*$.

```
FullSimplify[Reduce[(b + (-1 + a) m) (2 a^2 + b + b^2 - 2 a (1 + b (2 + m))) < 0, a],
 Assumptions -> {0 < a < b < 1, a + b < 1, 0 < r < 0.5`, 0 < m < 1}]
```

$$(b (2 + m (4 + 3 m)) < m (2 + m (3 + 2 m)) \&\& \left(2 a + \sqrt{1 + b (2 (1 + m) + b (2 + m (4 + m)))} < 1 + b (2 + m) \,||\, b + a m > m\right)) \,||\, (m (2 + m (3 + 2 m)) == b (2 + m (4 + 3 m)) \&\& b + a m \neq m) \,||\, \left(b (2 + m (4 + 3 m)) > m (2 + m (3 + 2 m)) \&\& \left(b + a m < m \,||\, 2 a + \sqrt{1 + b (2 (1 + m) + b (2 + m (4 + m)))} > 1 + b (2 + m)\right)\right)$$

```
FullSimplify[Solve[(b + (-1 + a) m) (2 a^2 + b + b^2 - 2 a (1 + b (2 + m))) == 0, a],
 Assumptions → {0 < a < b < 1, a + b < 1, 0 < r < 0.5`, 0 < m < 1}]
```

$$\left\{\left\{a \to 1 - \frac{b}{m}\right\}, \left\{a \to \frac{1}{2} \left(1 + b (2 + m) - \sqrt{1 + 2 b (1 + m) + b^2 (2 + m (4 + m))}\right)\right\}, \left\{a \to \frac{1}{2} \left(1 + b (2 + m) + \sqrt{1 + 2 b (1 + m) + b^2 (2 + m (4 + m))}\right)\right\}\right\}$$

```
FullSimplify[Reduce[0 < 1/2 (1 + b (2 + m) - Sqrt[1 + 2 b (1 + m) + b^2 (2 + m (4 + m))]) < 1],
 Assumptions → {0 < a < b < 1, a + b < 1, 0 < r < 0.5`, 0 < m < 1}]
```

True

```
FullSimplify[Reduce[0 < 1/2 (1 + b (2 + m) + Sqrt[1 + 2 b (1 + m) + b^2 (2 + m (4 + m))]) < 1],
 Assumptions → {0 < a < b < 1, a + b < 1, 0 < r < 0.5`, 0 < m < 1}]
```

False

This suggests that the first value is the one we are interested in, *i.e.*

$$a^* = \frac{1}{2}\left[1 + b\,(2 + m) - \sqrt{1 + 2\,b\,(1 + m) + b^2\,[2 + m\,(4 + m)]}\,\right] \tag{32}$$

```
critValA := 1/2 (1 + b (2 + m) - Sqrt[1 + 2 b (1 + m) + b^2 (2 + m (4 + m))])

Plot[{critValA /. {m → 0.1}, critValA /. {m → 0.2}, critValA /. {m → 0.3},
   critValA /. {m → 0.4}, critValA /. {m → 0.5}, critValA /. {m → 0.6},
   critValA /. {m → 0.7}, critValA /. {m → 0.8}, critValA /. {m → 0.9}, critValA /. {m → 1}},
  {b, 0, 1}, PlotRange → {{0, 1}, {0, 0.5}}, AspectRatio → Automatic,
 AxesLabel → {b, a}, PlotLabel → "Critical a  for strictly-positive optimum r"]
```



Plots of the approximate condition for $r_{opt} \neq 0$

From now on, we denote by r_{opt} the recombination rate at which the average invasion probability $\overline{\pi}(r)$ of A_1 is maximised.

As a function of *a* and *b* for a given migration rate *m*

Theory

We first determine the contour lines:

```
Simplify[Reduce[condApprox1Func[a, b, m] == 0, {a, b}], Assumptions → assumeGeneral]
```

$$b + a\,m == m \;||\; 2\,a\,(2 + m) + \sqrt{1 - 4\,a\,m + 4\,a^2\,(2 + 4\,m + m^2)} == 1 + 2\,b$$

We recognise that the first condition is equivalent to the definition of the critical migration rate for the stability of the marginal one-locus migration-selection equilibrium at the B-locus, $m_{\text{crit},2} = \frac{b}{1-a}$. The second condition defines a new critical migration rate that distinguishes between the regimes $r_{\text{opt}} > 0$ and $r_{\text{opt}} = 0$. This is given by

```
mCritRRule = Solve[2 a (2 + m) + Sqrt[1 - 4 a m + 4 a^2 (2 + 4 m + m^2)] == 1 + 2 b, m] // FullSimplify
```

$$\left\{\left\{m \to \frac{2\,(-1 + a)\,a + b - 4\,a\,b + b^2}{2\,a\,b}\right\}\right\}$$

$$m_{r_{\text{opt}}} = \frac{b - 2\,(1 - a)\,a - 4\,a\,b + b^2}{2\,a\,b} \tag{33}$$

In the plots below, the cyan and yellow contour lines both correspond to the case where $\bar{\pi}'(r = 0) = 0$. As a function of a (and m), the cyan line is given by $b(a, m) = m\,(1 - a)$ and the yellow line by

$$b\,(a,\ m) = \frac{1}{2}\left(-1 + 4\,a + 2\,a\,m + \sqrt{1 + 8\,a^2 - 4\,a\,m + 16\,a^2\,m + 4\,a^2\,m^2}\right) \tag{34}$$

```
Solve[b + a m == m, b]
```

$$\{\{b \to m - a\,m\}\}$$

```
Solve[2 a (2 + m) + Sqrt[1 - 4 a m + 4 a^2 (2 + 4 m + m^2)] == 1 + 2 b, b]
```

$$\left\{\left\{b \to \frac{1}{2}\left(-1 + 4\,a + 2\,a\,m + \sqrt{1 + 8\,a^2 - 4\,a\,m + 16\,a^2\,m + 4\,a^2\,m^2}\right)\right\}\right\}$$

```
contour1[a_, m_] := m (1 - a);
contour2[a_, m_] := 1/2 (-1 + 4 a + 2 a m + Sqrt[1 + 8 a^2 - 4 a m + 16 a^2 m + 4 a^2 m^2]);
```

Alternatively, we can express the contour line as a function of b and m:

```
Solve[b + a m == m, a]
```

$$\left\{\left\{a \to \frac{-b + m}{m}\right\}\right\}$$

```
Solve[2 a (2 + m) + Sqrt[1 - 4 a m + 4 a^2 (2 + 4 m + m^2)] == 1 + 2 b, a]
```

$$\left\{\left\{a \to \frac{1}{4}\left(2 + 4\,b + 2\,b\,m - \sqrt{8\,(-1 - b)\,b + (-2 - 4\,b - 2\,b\,m)^2}\right)\right\},\right.$$
$$\left.\left\{a \to \frac{1}{4}\left(2 + 4\,b + 2\,b\,m + \sqrt{8\,(-1 - b)\,b + (-2 - 4\,b - 2\,b\,m)^2}\right)\right\}\right\}$$

where only the first one is biologically valid ($0 < a < 1$).

```
contour1Alt[b_, m_] := 1 - b / m;
contour2Alt[b_, m_] := 1/4 (2 + 4 b + 2 b m - Sqrt[8 (-1 - b) b + (-2 - 4 b - 2 b m)^2]);
```

In the plots, the red line marks $a = b$ (recall that we are interested in cases where $a < b$). Dark grey shading represents the parameter space where, approximately, $\bar{\pi}'(r = 0) > 0 \Rightarrow r_{\text{opt}} > 0$, and lighter grey shading represents the range where, approximately, $\bar{\pi}'(r) < 0 \Rightarrow r_{\text{opt}} = 0$.

```
condApprox1Func[aa_, bb_, mm_] :=
  Chop[(b + (-1 + a) m) (2 a^2 + b (1 + b) - 2 a (1 + b (2 + m))) /. {a → aa, b → bb, m → mm}]
```

Remarks to the plots below:

- The red line corresponds to $a = b$. We are only interested in the case $a < b$, which corresponds to the area below the red line.
- The cyan line corresponds to $a = (m - b)/m = 1 - b/m$, which corresponds to the critical migration rate $m_{\text{crit},2} = \frac{b}{1-a}$. The marginal one-locus migration-selection equilibrium E_B is exists if and only if $m < m_{\text{crit},2}$. Existence of E_B is a necessary condition for the invasion of the A_1 mutant via E_B. The area in the plots above the cyan line corresponds to the case $m < m_{\text{crit},2}$.
- The yellow line has the following meaning: Whenever we are above the cyan line, the yellow line separates the parameter space into a parameter sub-space where $\overline{\pi}'(0) < 0 \Rightarrow r_{\text{opt}} = 0$ (dark grey, to the left of the yellow line) and another one where $\overline{\pi}'(0) > 0 \Rightarrow r_{\text{opt}} > 0$ (dark purple, to the right of the yellow line).

```
FullSimplify[a /. Solve[m == b / (1 - a), a]]
```

$\left\{1 - \frac{b}{m}\right\}$

```
FullSimplify[Reduce[m < b / (1 - a), a], Assumptions → {0 < a < b < 1, a + b < 1, 0 < m < 1}]
```

b + a m > m

$$\texttt{mCrit5} := \frac{a\,(b - a + r)}{(a - r)\,(a - b) + r\,(1 - a)}$$

```
mCrit5 /. {r → 0} // FullSimplify
```

-1

- Plot

```
myM = 0.3; (*0.3; 0.032; 0.5;*)
myBmin = 0.;
myBmax = .6;(*0.6;0.045;1.;*)
myACoord = 2;(*0.025;2;*) (* To show a point of interest,
choose coordinates that are within the plot range *)
myBCoord = 2;(*0.04;2;*)(* To show a point of interest,
choose coordinates that are within the plot range *)
myAmin = myBmin(* Do not change this. *);
myAmax = myBmax (* Do not change this. *);
plotOptRecombRate = Show[ContourPlot[condApprox1Func[a, b, myM],
    {b, myBmin, myBmax}, {a, myAmin, myAmax}, Contours → {-100, 0, 100},
    ContourShading → {Red, RGBColor[0.25, 0.25, 0.25], RGBColor[0.6, 0.6, 0.6], Blue},
    FrameLabel → {"Selection coefficient b", "Selection coefficient a"},
    PlotRange → {{myBmin, myBmax}, {myBmin, myBmax}, Full},
    FrameTicksStyle → Directive[Medium, Black, FontSize → 16],
    LabelStyle → {Directive[FontSize → 20], FontFamily → "Helvetica"}, PlotPoints → 60],
   Plot[contour1Alt[b, myM], {b, myBmin, myBmax}, PlotStyle → {Thick, Cyan},
    Filling → Bottom, FillingStyle → Directive[RGBColor[0.9, 0.9, 0.9], Opacity[1]]],
   Plot[contour2Alt[b, myM], {b, myBmin, myBmax}, PlotStyle → {Thick, RGBColor[1, 1, 0]}],
   Plot[b, {b, myBmin, myBmax}, PlotStyle → {Thick, Red}, Filling → Top,
    FillingStyle → Directive[RGBColor[0.98, 0.98, 0.98], Opacity[1]]],
   ListPlot[{{myBCoord, myACoord}}, PlotStyle → {RGBColor[0, 1, 0, 0.65]}]]
   (* Dark: critVal < 0 ⇔ π̄'[0] > 0 → r_opt ≠ 0;
 Bright: critVal > 0 ⇔ π̄'[0] < 0  → r_opt = 0; *)
```



Figure 1: Classification of the behaviour of the average invasion probability as a function of the recombination rate, $\bar{\pi}(r)$. The dark grey area indicates where the derivative of $\bar{\pi}(r)$ with respect to r, evalutated at $r = 0$, is positive ($\bar{\pi}'(0) > 0$) and the optimal recombination rate is therefore positive ($r_{opt} > 0$). The medium grey area shows the parameter range for which $\bar{\pi}'(0) \leq 0$ and therefore $r_{opt} = 0$. Together, these two areas indicate where A_1 can invade via the marginal one-locus migration-selection equilibrium E_B if r is sufficiently small. The light grey area shows where E_B does not exist and A_1 cannot invade. Finally, the area above $a = b$ is not of interest, as we focus on mutations that are weakly beneficial compared to selection at the background locus ($a < b$). The migration rate is $m = 0.3$.

```
Export[figPath <> "plotOptRecombRate.tiff",
 Rasterize[plotOptRecombRate, ImageResolution → 72], "TIFF"]
```

```
/Users/Simon/Documents/LocAdD/results/130606/nonZeroOptRecRate/figures/
  plotOptRecombRate.tiff
```

Understanding the dependence of π_1° on m

We observed that π_1° increases with the migration rate m, which is perhaps counterintuitive. Because π_1° is essentially determined by the ratio of the marginal fitness w_1 of $A_1 B_1$ to the mean resident fitness $\overline{w}$, we investigate the dependence of w_1 and $\overline{w}$ on m.

```
w1Rule
```

w1 → qEqB w13 + (1 - qEqB) w14

```
wBarRule
```

wBar → qEqB^2 w33 + 2 (1 - qEqB) qEqB w34 + (1 - qEqB)^2 w44

```
qEqBRule
```

$$\left\{qB \to \frac{w34 - m\, w34 - wBarTilde}{(-1+m)\,(w33 - w34)}\right\}$$

```
w13AddRule := w13 → 1 + b;
w14AddRule := w14 → 1;
w33AddRule := w33 → 1 - a + b;
w34AddRule := w34 → 1 - a;
w44AddRule := w44 → 1 - a - b;
addFitRule := {w13AddRule, w14AddRule, w33AddRule, w34AddRule, w44AddRule}
```

qEqBAddRule := qEqB → $\frac{b - m\,(1-a)}{b\,(1+m)}$

```
w1Add = w1 /. w1Rule /. addFitRule /. qEqBAddRule // Simplify
```

$$\frac{1+b+a\,m}{1+m}$$

```
wBarAdd = wBar /. wBarRule /. addFitRule /. qEqBAddRule // FullSimplify
```

$$\frac{(-1+a-b)\,(-1+m)}{1+m}$$

wRatioAdd = $\frac{\text{w1Add}}{\text{wBarAdd}}$ // FullSimplify

$$\frac{1+b+a\,m}{(-1+a-b)\,(-1+m)}$$

w1AddFunc[m_] := $\frac{1+b+a\,m}{1+m}$

wBarAddFunc[m_] := $\frac{(1-a+b)\,(1-m)}{1+m}$

wRatioAddFunc[m_] := $\frac{1+b+a\,m}{(1-a+b)\,(1-m)}$

```
Manipulate[Plot[{w1AddFunc[m] /. {a → mya, b → myb},
    wBarAddFunc[m] /. {a → mya, b → myb}, wRatioAddFunc[m] /. {a → mya, b → myb}},
   {m, 0, 1}, PlotStyle → {{RGBColor[0.6, 0.6, 0.9]}, {Black}, {Black, Thick, Dashed}},
   Frame → True, FrameStyle → {{Black, Opacity[0]}, {Black, Opacity[0]}},
   FrameLabel → {"Migration rate m", "Fitness"},
   LabelStyle → {Directive[FontSize → 14], FontFamily → "Helvetica"}],
  {{mya, 0.02}, 0, 1}, {{myb, 0.02}, 0, 1}]
```



General rules and assumptions

```
ruleSmallForces := {a → α ϵ, b → β ϵ, m → μ ϵ, r → ρ ϵ}
ruleReturnOrigin := {α → a / ϵ, β → b / ϵ, μ → m / ϵ, ρ → r / ϵ}

ruleWeakMigration := {m → μ ϵ}

assumeGeneral := {0 < a < b < 1, a + b < 1, 0 < r < 0.5, 0 < m < 1}
```

qEqRule := qEq → $\frac{b - m + a\,m}{b\,(1 + m)}$ (* The frequency of the B_1 allele at the marginal one-locus migration-selection equilibrium *)

Functions

This function solves the system of transcendental equations obtained with the two-type branching process numerically. For a derivation, see *Mathematica* notebook '2LocContIsland_Stoch_Discr.nb'.

```
probEstablAMApproxPolymContFunc::usage =
  "probEstablAMApproxPolymContFunc[r, m1, a1, b1, γ111, γ121, γ211, γ221, qC]";
probEstablAMApproxPolymContFunc[r_, m1_, a1_, b1_, γ111_, γ121_, γ211_, γ221_, qC_] :=
  Module[{qEq, wbar, w1, w2, w14, λ1, pgf1, pgf2, qSol},
   qEq = 1/(2 * b1 * (1 + m1)) (b1 - m1 + a1 * m1 + 2 * b1 * m1 * qC +
        √(-4 * b1 * (-1 + a1 + b1) * m1 * (1 + m1) * qC + (b1 + (-1 + a1) * m1 + 2 * b1 * m1 * qC)^2));
   (* See 120820_twoLocusContinentIslandDiscreteDetPolyCont.nb *)
wbar = 1 - a1 + b1 * (-1 + 2 * qEq);
   w1 = 1 + b1 * qEq + (-1 + qEq) * γ111;
   w2 = 1 + b1 * (-1 + qEq) - qEq * γ111 + γ121 * (-1 + qEq);
   w14 = 1 - γ111;
   (* Leading eigenvalue of the mean matrix; Note that q_c does *not* enter here! *)
   λ1 = -1/(2 wbar) (-1 + m1) *
     (w1 - r * w14 + w2 + (w1^2 + r^2 * w14^2 + w1 * (2 * (-1 + 2 * qEq) * r * w14 - 2 * w2) +
          2 * (1 - 2 * qEq) * r * w14 * w2 + w2^2)^(1/2));
   (* Probability generating functions *)
   pgf1[s1_, s2_] := Exp[
     -(r * (1 - m1) * (1 - qEq) * (1 - s2) * w14)/wbar - ((1 - m1) * (1 - s1) * (w1 - r * (1 - qEq) * w14))/wbar];
   pgf2[s1_, s2_] := Exp[-(r * (1 - m1) * qEq * (1 - s1) * w14)/wbar -
     ((1 - m1) * (1 - s2) * (-r * qEq * w14 + w2))/wbar];
   qSol = FindRoot[{pgf1[q1, q2] == q1, pgf2[q1, q2] == q2}, {q1, 0.5}, {q2, 0.5}];
   (* Return the probability of establishment, 1-q *)
Return[{λ1, (1 - q1), (1 - q2), qEq * (1 - q1) + (1 - qEq) * (1 - q2), qEq} /. qSol]
  ];
```

Checks

- **Using eight constants**

```
b + b^2 + a b m == b (1 + b + a m) // Simplify

True

(a - 1) m r - b m r == -m (1 - a + b) r // Simplify

True

(1 - a) m r + b m r == m (1 - a + b) r // Simplify

True

b r + (a - 1) m r == (b - (1 - a) m) r // Simplify

True

b + a b m - b^2 m == b (1 + m (a - b)) // Simplify

True

-b r + (1 - a) m r == ((1 - a) m - b) r // Simplify

True
```

```
{(A + B r) s1 + C r s2 - D /. Flatten[{ruleA, ruleB, ruleC, ruleD}]} ==
  {((b (1 + b + a m) - m (1 - a + b) r) s1 + (m (1 - a + b) r) s2 - b (1 + b + a m)) / (b (1 - a + b))} //
 FullSimplify
```

True

```
{E r s1 + (F + G r) s2 - H /. Flatten[{ruleE, ruleF, ruleG, ruleH}]} ==
  {(((b - (1 - a) m) r) s1 + (b (1 + m (a - b)) - (b - (1 - a) m) r) s2 - b (1 + m (a - b))) /
    (b (1 - a + b))} // FullSimplify
```

True

■ Using four constants (E, F, H, J)

```
myA = (1 - m) (w13 qhat + w14 (1 - qhat) (1 - r)) / wbar;
myB = (1 - m) r w14 qhat / wbar;
myC = (1 - m) r w14 (1 - qhat) / wbar;
myD = (1 - m) (w24 (1 - qhat) + w14 qhat (1 - r)) / wbar;
```

Additive Fitnesses

```
w14Rule = w14 → 1;
```

$$\text{qhatRule} = \text{qhat} \to \frac{b - m\,(1 - a)}{b\,(1 + m)};$$

```
w33Rule = w33 → 1 - a + b;
w34Rule = w34 → 1 - a;
w44Rule = w44 → 1 - a - b;
w13Rule = w13 → 1 + b;
w24Rule = w24 → 1 - b;
```

$$\text{wbarRule} = \text{wbar} \to \text{qhat}^2\,\text{w33} + 2\,\text{qhat}\,(1 - \text{qhat})\,\text{w34} + (1 - \text{qhat})^2\,\text{w44} \;/.\; \text{qhatRule} \;/.\; \text{w33Rule} \;/.\; \text{w34Rule} \;/.\; \text{w44Rule};$$

```
lambda11Add = myA /. w13Rule /. w14Rule /. qhatRule /. wbarRule // FullSimplify
```

$$\frac{1 + b + a\,m}{1 - a + b} - \frac{m\,r}{b}$$

```
lambda21Add = myB /. w14Rule /. qhatRule /. wbarRule // FullSimplify
```

$$\frac{(b + (-1 + a)\,m)\,r}{b\,(1 - a + b)}$$

```
lambda12Add = myC /. w14Rule /. qhatRule /. wbarRule // FullSimplify
```

$$\frac{m\,r}{b}$$

```
lambda22Add = myD /. w24Rule /. w14Rule /. qhatRule /. wbarRule // FullSimplify
```

$$\frac{b + a\,b\,m - b^2\,m - b\,r + m\,r - a\,m\,r}{b - a\,b + b^2}$$

```
ARule = myA → lambda11Add;
BRule = myB → lambda21Add;
CRule = myC → lambda12Add;
DRule = myD → lambda22Add;
```

Assuming small evolutionary forces

```
assumeSmallForces := {a → α ϵ, b → β ϵ, m → μ ϵ, r → ρ ϵ}
```

```
resubst := {α → a / ϵ, β → b / ϵ, μ → m / ϵ, ρ → r / ϵ}
```

```
Series[{{lambda11Add, lambda12Add}, {lambda21Add, lambda22Add}} /. assumeSmallForces,
   {ϵ, 0, 1}] /. resubst // Normal // MatrixForm
```

$$\begin{pmatrix} 1 + a - \frac{m\,r}{b} & \frac{m\,r}{b} \\ r - \frac{m\,r}{b} & 1 + a - b - r + \frac{m\,r}{b} \end{pmatrix}$$

$$\text{myE} = \frac{1 + b + a\,m}{1 - a + b};$$

$$\text{myF} = -\frac{m}{b};$$

$$\text{myH} = \frac{b - (1 - a)\,m}{b\,(1 - a + b)};$$

$$\text{myJ} = \frac{1 + m\,(a - b)}{1 - a + b};$$

$$\text{pgf1}[\text{s1\_}, \text{s2\_}] := e^{-\text{myA}\,(1-\text{s1}) - \text{myC}\,(1-\text{s2})}$$

$$\text{pgf2}[\text{s1\_}, \text{s2\_}] := e^{-\text{myB}\,(1-\text{s1}) - \text{myD}\,(1-\text{s2})}$$

pgf1[s1, s2] /. ARule /. CRule

$$e^{-\left(\frac{1+b+a\,m}{1-a+b} - \frac{m\,r}{b}\right)\,(1-\text{s1}) - \frac{m\,r\,(1-\text{s2})}{b}}$$

pgf2[s1, s2] /. BRule /. DRule

$$e^{-\frac{(b+(-1+a)\,m)\,r\,(1-\text{s1})}{b\,(1-a+b)} - \frac{\left(b+a\,b\,m-b^2\,m-b\,r+m\,r-a\,m\,r\right)\,(1-\text{s2})}{b-a\,b+b^2}}$$

$$((b\,(1 + b + a\,m) - m\,(1 - a + b)\,r)\,\text{s1} + (m\,(1 - a + b)\,r)\,\text{s2} - b\,(1 + b + a\,m)) \,/\, (b\,(1 - a + b)) == -\left(\frac{1 + b + a\,m}{1 - a + b} - \frac{m\,r}{b}\right)(1 - \text{s1}) - \frac{m\,r\,(1 - \text{s2})}{b} \;//\; \text{FullSimplify}$$

True

$$-\frac{(b + (-1 + a)\,m)\,r\,(1 - \text{s1})}{b\,(1 - a + b)} - \frac{\left(b + a\,b\,m - b^2\,m - b\,r + m\,r - a\,m\,r\right)(1 - \text{s2})}{b - a\,b + b^2} == (((b - (1 - a)\,m)\,r)\,\text{s1} + (b\,(1 + m\,(a - b)) - (b - (1 - a)\,m)\,r)\,\text{s2} - b\,(1 + m\,(a - b)))\,/\,(b\,(1 - a + b)) \;//\; \text{FullSimplify}$$

True

$$\text{Collect}\left[-\left(\frac{1 + b + a\,m}{1 - a + b} - \frac{m\,r}{b}\right)(1 - \text{s1}) - \frac{m\,r\,(1 - \text{s2})}{b} \;/.\; r \to 0,\; \{1 - \text{s1}\}\right]$$

$$-\frac{(1 + b + a\,m)\,(1 - \text{s1})}{1 - a + b}$$

$$\text{Collect}\left[-\frac{(b + (-1 + a)\,m)\,r\,(1 - \text{s1})}{b\,(1 - a + b)} - \frac{\left(b + a\,b\,m - b^2\,m - b\,r + m\,r - a\,m\,r\right)(1 - \text{s2})}{b - a\,b + b^2} \;/.\; r \to 0,\; \{1 - \text{s1}\}\right]$$

$$-\frac{\left(b + a\,b\,m - b^2\,m\right)(1 - \text{s2})}{b - a\,b + b^2}$$

$$\text{pgfAdd1}[\text{s1\_}, \text{s2\_}] = e^{(\text{myE}+\text{myF}\,r)\,\text{s1} - \text{myF}\,r\,\text{s2} - \text{myE}}$$

$$e^{-\frac{1+b+a\,m}{1-a+b} + \left(\frac{1+b+a\,m}{1-a+b} - \frac{m\,r}{b}\right)\,\text{s1} + \frac{m\,r\,\text{s2}}{b}}$$

$$\text{pgfAdd2}[\text{s1\_}, \text{s2\_}] = e^{\text{myH}\,r\,\text{s1} + (\text{myJ} - \text{myH}\,r)\,\text{s2} - \text{myJ}}$$

$$e^{-\frac{1+(a-b)\,m}{1-a+b} + \frac{(b-(1-a)\,m)\,r\,\text{s1}}{b\,(1-a+b)} + \left(\frac{1+(a-b)\,m}{1-a+b} - \frac{(b-(1-a)\,m)\,r}{b\,(1-a+b)}\right)\,\text{s2}}$$

$$-\frac{1 + b + a\,m}{1 - a + b} + \left(\frac{1 + b + a\,m}{1 - a + b} - \frac{m\,r}{b}\right)\text{s1} + \frac{m\,r\,\text{s2}}{b} == -\left(\frac{1 + b + a\,m}{1 - a + b} - \frac{m\,r}{b}\right)(1 - \text{s1}) - \frac{m\,r\,(1 - \text{s2})}{b} \;//\; \text{FullSimplify}$$

True

$$-\frac{1+(a-b)\,m}{1-a+b}+\frac{(b-(1-a)\,m)\,r\,s1}{b\,(1-a+b)}+\left(\frac{1+(a-b)\,m}{1-a+b}-\frac{(b-(1-a)\,m)\,r}{b\,(1-a+b)}\right)s2 ==$$
$$-\frac{(b+(-1+a)\,m)\,r\,(1-s1)}{b\,(1-a+b)}-\frac{\left(b+a\,b\,m-b^2\,m-b\,r+m\,r-a\,m\,r\right)(1-s2)}{b-a\,b+b^2}\ //\ \text{FullSimplify}$$

```
True
```

Polymorphic continent with additive fitnesses

Not shown in detail here.

Implementation

Derivatives of $f_i(s_1, s_2)$ for $r > 0$ but small

Assuming all evolutionary forces to be small